

UNITED STATES DEPARTMENT OF THE INTERIOR
GEOLOGICAL SURVEY

SIROTEM II IEEE 488-1978 interface and
controlling software

By

Jerry Bradley and Paul Raab

Open-File Report 83-249

1983

This report is preliminary and has not been reviewed for conformity with U.S. Geological Survey editorial standards. Any use of trade names is for descriptive purposes only and does not imply endorsement by the USGS. The program presented herein was written in BASIC for a Hewlett-Packard 85 computer. Although program tests have been made, no guarantee (expressed or implied) is made by the authors regarding program correctness, accuracy, or proper execution on all computer systems.

Introduction

Geophysical data is often stored on magnetic tape, or a similar medium, as it is collected in the field. If this data is transferred directly to a computer via an electrical interface, processing time is reduced and typing errors eliminated. This report presents the logic diagrams and controlling software for an 8-bit parallel interface which is compatible with the Australian GEOEX SIROTEM, a time domain electromagnetic (TDEM) data collection system.

The SIROTEM measures and averages the transient earth response to a transmitted current pulse over 32 discrete time intervals. The resultant data is stored on magnetic tape. In the unmodified SIROTEM system, stored data can only be output to an internal printer (GEOEX SIROTEM II operation manual, 1980). With the interface described in this paper, data can be transferred from the SIROTEM to a computer with IEEE Standard 488-1978 compatible hardware. The software to control this interface card has been written in BASIC for a Hewlett-Packard 85 desktop computer equipped with the Plotter/Printer, Input/Output, Advanced Programming, and Matrix ROM's and Hewlett-Packard's implementation of the IEEE 488-1978 standard interface bus (the HP-IB).

SIROTEM to HP-85 Interface Logic

The IEEE 488-1978 compatible SIROTEM interface enables the transfer of data from the SIROTEM tape cassette to a remote computer and also provides for control of all SIROTEM tape movement functions. A mix of TTL and CMOS logic was used in the design of this interface. Power for the interface is provided by the SIROTEM +5 volt digital supply; the standby current drain is approximately 12 milliamperes. Figure 1 illustrates the tape motion control logic and the logic that initiates a data transfer from the SIROTEM tape cassette. Figure 2 depicts the logic used for temporary storage and formatting of SIROTEM data before it is transferred over the interface. The logic circuits in Figure 3 implement the fast handshake signal \overline{DAV} (data available), the end of data on tape (EOI) signal (generated when blank tape is detected), and the logic used to inhibit the SIROTEM internal printer during a data transfer.

When an IEEE 488-1978 standard interface is connected from an external controller, such as the HP-85, to the SIROTEM, a hardware jumper (Figure 1, PRINT INH) disables the SIROTEM's internal printer and establishes the interface as the output device. A SIROTEM tape read cycle is started through the use of the HP-IB control lines REN (remote enable) and ATN (attention). While ATN is held

high, REN is set low and then pulsed high. This sequence generates an INIT (initialize) signal into the interface (Figure 1) and instructs the SIROTEM microprocessor to transfer one file from the magnetic tape.

Twelve bit parallel data representing three BCD digits are read from the SIROTEM microprocessor (Figure 2, BDX0-BDX11) to a series of parallel loaded, up/down counters (Figure 2, IC's 10-12) for temporary storage. Since each BCD digit has a superfluous printer control 2⁰ bit added to it by the SIROTEM software, each counter is decremented once to remove this bit. Data strobes (Figure 2, DS3-DS7) are used to enter five sets of three BCD digits, which comprise one line of the SIROTEM printer output, into the storage devices for transfer over the interface. Next, the DAV control line on the interface is used to signal the computer that 8 bits of data are present on the data bus. After reading the bus, the computer sends the control NDAC (not data accepted) to the interface. This signal causes the next 8 bits of data to be formatted and reinitiates DAV to start the cycle over again. The timing diagram of Figure 2 shows the sequence of DAV and NDAC required to format three BCD digits (4 bits each) of the SIROTEM into two 8 bit words for transfer to the computer. This sequence is repeated until either all data has been read from the SIROTEM tape or the computer stops sending the INIT signal.

The logic circuits to control the SIROTEM tape positioning functions are shown in Figure 1. Binary codes for each of the tape functions REWIND, ADVANCE, and REVERSE are sent on the HP-IB interface data lines to the SIROTEM interface by the controlling computer. These codes are designated as command sequences by setting the control lines ATN and REN low and then pulsing REN high.

SIROTEM to HP-85 Interface Software

The program "RDTAPE" (Appendix I) provides the software commands necessary to control the SIROTEM to HP-85 interface card. This program provides for positioning of the SIROTEM tape via commands to rewind, advance, and reverse the tape. It also enables the transfer of data from the SIROTEM tape to the HP-85 computer for error checks, decoding, formatting, and output to an HP-85 tape for later processing.

The transfer of data from the SIROTEM magnetic tape to the HP-85 is initiated by setting the HP-IB control line REN low with ATN high and then pulsing REN high. This sequence instructs the SIROTEM microprocessor to transfer one file from the magnetic tape. A fast handshake transfer is used to enter this data into HP-85 input buffers. Data

is checked for errors in transmission. If an error is found, the SIROTEM tape is reversed and read again (Note: if transmission errors occur more than four times for any given file, that file will be skipped).

The number of data records stored on each SIROTEM tape file depends on the number of time intervals selected (see GEOEX SIROTEM II operation manual, 1980, p. 62, for a description of the SIROTEM tape format). An end of file condition is recognized by a sequence of three spaces in the tape record containing the number of stacks and transmitter loop current. When this sequence is found, program control is passed to a routine to decode and format the data for output to an HP-85 tape. An end of data on tape condition will be recognized by the termination of the fast handshake transfer with EOI (an end or identify control line on the HP-IB) instead of a buffer full condition. The SIROTEM to HP-85 interface will generate an EOI when there is no valid data on the SIROTEM tape, a situation which occurs when blank tape is found. If a SIROTEM tape has been overwritten, there may still be old data records on tape past the point where new data is stored. In this case the program will keep on reading the tape until a blank area is reached. Occasional checking of the program's progress or erasing SIROTEM tapes before field use should obviate the need for a more sophisticated recognition of an end of valid data on tape condition.

After a file has been read, the input data are decoded and further error checking occurs. If formatting errors are found, the SIROTEM tape is reversed and read again. It should be noted that formatting errors may be due to errors in transmission or to problems with the data as stored on the SIROTEM tape. The program will attempt to re-read a SIROTEM tape file a maximum of four times. Processing will continue if no errors are found or if the tape file is re-read more than four times. At this step, the SIROTEM data is displayed and output to tape.

The lack of a full handshake capability in the SIROTEM's microprocessor and the speed at which data words are output from the SIROTEM to HP-85 interface requires the use of the fast handshake transfer instead of an interrupt transfer. This can cause problems if the fast handshake transfer is not terminated properly, in which case the HP-85 will "hang-up" and the keyboard will not function. If such a problem occurs, the HP-85 should be turned off and processing restarted.

Before connecting the interface, both the SIROTEM and the HP-85 should be off. After loading and starting the program, a cautionary note is displayed on the HP-85 CRT to

make sure that SIROTEM data tapes are write protected before being read. Then the user is queried for a project title (32 characters maximum) and a date (8 characters maximum). This information is stored along with the SIROTEM data. Next, the special function keys of the HP-85 are used to allow the selection of program options. When pressed, these keys (as labeled on the HP-85 CRT) will execute the following subroutines:

REW - outputs a command to rewind the SIROTEM tape. A programmed wait of 10 seconds is used to allow this operation to complete.

FWD - outputs a command to advance the SIROTEM tape one file. A programmed wait of 5 seconds is used to allow this operation to complete.

REV - outputs a command to reverse the SIROTEM tape one file. A programmed wait of 5 seconds is used to allow this operation to complete.

READ - initiates the subroutine which will read and process data from the SIROTEM tape. The user can read several or all files on the SIROTEM tape. Also, there is an option to store the data on an HP-85 tape for future processing. If data is stored, a general file name must be entered. Output file names will consist of this name and an appended file number (see Figure 4 for HP-85 storage format). TDEM data files created by the program "RDTAPE" can be processed and apparent resistivities calculated and displayed using programs detailed in Raab and Frischknecht, 1983. Defaults for the read options are displayed and can be selected by pressing end line without entering a response. As each file is processed, the SIROTEM header and read information (e.g. errors noted, output file name) are printed on the HP-85 printer.

HELP - executes a subroutine which will print or display a brief help library explaining the basic operation of the program.

References

GEOEX PTY. LTD., 1980, GEOEX SIROTEM II transient EM system operation manual: Unley, S.A., Australia.

Raab, Paul and Frischknecht, Frank, 1983, Desktop computer processing of coincident and central loop time domain electromagnetic data: U.S. Geological Survey Open-File Report 83-240.




Figure 1. Tape motion control and data transfer initiation logic diagram.

Character Position	
1234567890123456789012	
Record Number	
1	ROTAPE TEST
2	4/13/83
3	# OF CHANNELS = 24
4	03+0783000003
5	1+1460 4 000
	2+4539 3 000
	3+1611 3 000
.	4+6830 2 000
	5+3180 2 000
	6+1447 2 000
	7+6869 1 000
.	8+4516 1 000
	9+2903 1 000
	10+2183 1 000
	11+1279 1 000
	12+6951 0 001
.	13+4352 0 001
	14+2675 0 001
	15+2157 0 001
	16+1002 0 002
	17+0320 0 003
	18+0094 0 006
.	19+0077 0 012
	20+0003 0 306-
	21+0024 0 041
	22+0123 0 005-
	23+0093 0 009-
4+N	24+0070 0 011
	00+2040 2.4

Header Records

Data Records

Stacks & Current

Figure 4. HP-85 tape storage format for SIROTEM data.

Appendix I. Listing of the program "RDTAPE".

```
10 ! ***** RDTAPE *****
20 ! * READS SIROTEM DATA      *
30 ! * TAPES OVER THE HP-IB.   *
40 ! * OPTION TO WRITE DATA   *
50 ! * TO HP-85 TAPE IN A      *
60 ! * STANDARD FORMAT. DATA  *
70 ! * PRINTED AS READ. ERROR  *
80 ! * TRAPS WHERE POSSIBLE.   *
90 ! *****
100 !
110 ! PROGRAM CODED BY PAUL RAAB
120 ! INTERFACE LOGIC DESIGN BY JERRY BRADLEY
130 ! US GEOLOGICAL SURVEY
140 ! BRANCH OF GEOPHYSICS
150 ! MODIFIED 4/13/83
160 !
170 OPTION BASE 1
180 CRT IS 1 @ PRINTER IS 2
190 CLEAR
200 DIM A$(32),B$(8),C$(18),IO$(19),I1$(18),T$(400)
210 DIM O$(3200),O1$(20),O2$(850),Z$(32),NO$(10),N$(17)
220 SARRAY O2$
230 IOBUFFER IO$ @ IOBUFFER I1$
240 EO=0
250 !
260 ! ENTER HEADER
270 !
280 DISP "SIROTEM TAPE READING PROGRAM"
290 DISP @ DISP "WARNING : DATA TAPES SHOULD BE"
300 DISP "WRITE PROTECTED BEFORE BEING "
310 DISP "READ WITH THIS PROGRAM!!! "
320 DISP @ DISP "(PRESS CONTINUE)" @ PAUSE
330 CLEAR
340 DISP "ENTER GENERAL PROJECT TITLE      32 char max"
350 INPUT A$
360 DISP
370 DISP "ENTER DATE (e.g. 2/12/82) 8 max" @ INPUT B$
380 !
390 ! OPTION SELECT
400 !
410 ON KEY# 1,"REW" GOSUB 510
420 ON KEY# 2,"FWD" GOSUB 620
430 ON KEY# 3,"REV" GOSUB 730
440 ON KEY# 4,"READ" GOSUB 890
450 ON KEY# 5,"HELP" GOSUB 2530
460 SEND 7 ; MTA UNL LISTEN 7
470 CLEAR @ SFLAG 1 @ KEY LABEL
480 DISP "WAITING FOR KEY SELECTION"
490 IF FLAG(1) THEN 490 ELSE 470 ! WAIT FOR KEY HIT
```

```

500 !
510 ! REWIND THE TAPE
520 !
530 CLEAR @ CFLAG 1
540 DISP "WAIT FOR REWIND OPERATION"
550 SEND 7 ; CMD CHR$(4) ! REWIND COMMAND - BINARY 0100
560 ASSERT 7;80 ! SET ATN-REN LOW
570 ASSERT 7;16 ! PULSE REN HIGH
580 ASSERT 7;80 ! SET ATN-REN LOW
590 WAIT 10000
600 RETURN
610 !
620 ! ADVANCE THE TAPE
630 !
640 CLEAR @ CFLAG 1
650 DISP "WAIT FOR ADVANCE OPERATION"
660 SEND 7 ; CMD CHR$(8) ! FORWARD COMMAND - BINARY 1000
670 ASSERT 7;80 ! SET ATN-REN LOW
680 ASSERT 7;16 ! PULSE REN HIGH
690 ASSERT 7;80 ! SET ATN-REN LOW
700 WAIT 5000
710 RETURN
720 !
730 ! REVERSE THE TAPE
740 !
750 CLEAR @ CFLAG 1
760 IF EO<=4 THEN 810
770 PRINT
780 PRINT "READ ATTEMPTS HAVE FAILED"
790 PRINT "ON FILE NUMBER ";I;"!"
800 BEEP @ RETURN
810 DISP "WAIT FOR REVERSE OPERATION"
820 SEND 7 ; CMD CHR$(10) ! REVERSE COMMAND - BINARY 1010
830 ASSERT 7;80 ! SET ATN-REN LOW
840 ASSERT 7;16 ! PULSE REN HIGH
850 ASSERT 7;80 ! SET ATN-REN LOW
860 WAIT 5000
870 RETURN
880 !
890 ! READ THE TAPE
900 !
910 CLEAR @ CFLAG 1
920 DISP "AUTO READ? Y/N"
930 DISP "DEFAULT IS YES" @ INPUT Z$
940 IF Z$="" THEN IO=999 @ GOTO 1000
950 IF UPC$(Z$[1,1])#"N" THEN IO=1000 @ GOTO 1000
960 DISP @ DISP "ENTER NUMBER OF FILES TO BE READ"
970 DISP "DEFAULT = 1" @ INPUT Z$
980 IF Z$="" THEN IO=1 ELSE IO=VAL(Z$[1])
990 !
1000 ! STORE DATA OPTION
1010 !

```

```

1020 CLEAR
1030 N=1
1040 DISP "STORE DATA ON TAPE? Y/N"
1050 DISP "DEFAULT IS YES" @ INPUT Z$
1060 IF Z$="" THEN F=1 @ GOTO 1090
1070 IF UPC$(Z$[1,1])="N" THEN F=0 @ GOTO 1190 ELSE F=1
1080 !
1090 ! GET FILE NAME PARAMETERS
1100 !
1110 DISP @ DISP "ENTER GENERAL FILE NAME?" @ INPUT NO$
1120 DISP @ DISP "ENTER STARTING TAPE FILE NUMBER"
1130 DISP "NOTE:WILL BE APPENDED TO GENERAL"
1140 DISP "FILE NAME AND INCREMENTED BY 1"
1150 DISP "DEFAULT=1" @ INPUT Z$
1160 IF Z$#" " THEN N=VAL(Z$[1])
1170 IF N>1 THEN IO=N+IO-1
1180 !
1190 ! PROCESSING LOOP
1200 !
1210 FOR I=N TO IO
1220 EO,E1,E2=0 ! INITIALIZE ERROR COUNTERS
1230 !
1240 ! SIROTEM FHS TRANSFER
1250 !
1260 CLEAR
1270 IF EO>4 THEN 2370
1280 IO$="" @ I1$="" @ T$="" ! INITIALIZE INPUT STRINGS
1290 C,P0,P1=0 ! INITIALIZE POSITION COUNTERS
1300 SEND 7 ; MLA TALK 7
1310 ASSERT 7;64 @ ASSERT 7;0 @ ASSERT 7;64 ! PULSE REN
1320 TRANSFER 7 TO IO$ FHS ; COUNT 11 EOI
1330 IF LEN(IO$)<11 THEN 2450
1340 DISP IO$
1350 IF IO$[11,11]=CHR$(16) THEN 1370
1360 WAIT 5000 @ EO=EO+1 @ GOSUB 730 @ GOTO 1240
1370 T$=T$&IO$[2,11]
1380 TRANSFER 7 TO I1$ FHS ; COUNT 10 EOI
1390 DISP I1$
1400 IF I1$[10,10]=CHR$(16) THEN 1420
1410 WAIT 5000 @ EO=EO+1 @ GOSUB 730 @ GOTO 1240
1420 T$=T$&I1$
1430 IF I1$[5,6]=CHR$(17)&CHR$(16) THEN 1460 ! EOF
1440 I1$="" @ GOTO 1380
1450 !
1460 ! DECODE AND FORMAT DATA
1470 !
1480 CLEAR
1490 DISP "DECODING FILE ";I @ DISP
1500 P0=LEN(T$)
1510 FOR J=1 TO P0
1520 X=NUM(T$[J,J])
1530 X$=DTB$(X)

```

```

1540 P=(J-1)*8+1
1550 O$(P,P+7)=X$(9,16]
1560 NEXT J
1570 F$="11111111111111"
1580 P1=INT(P0/10)
1590 C$(1,16)="# OF CHANNELS = "
1600 C$(17,18)=VAL$(P1-2)
1610 FOR K=1 TO P0 STEP 10
1620 C=C+1 ! STRING COUNTER
1630 O1$="" ! INITIALIZE CONVERSION STRING
1640 FOR J=1 TO 10
1650 P=(K-1)*8+((J-1)*8+1)
1660 IF J=2 THEN O1$=O1$&"}"
1670 X=BTD(F$&O$(P,P+3])
1680 Y=BTD(F$&O$(P+4,P+7])
1690 X=BINCMP(X)
1700 Y=BINCMP(Y)
1710 IF X<10 THEN O1$=O1$&VAL$(X) @ GOTO 1780
1720 IF X=10 THEN O1$=O1$&"-" @ GOTO 1780
1730 IF X=11 THEN O1$=O1$&"." @ GOTO 1780
1740 IF X=12 THEN O1$=O1$&"9" @ GOTO 1780
1750 IF X=14 THEN O1$=O1$&" " @ GOTO 1780
1760 IF X=15 THEN O1$=O1$&"" @ GOTO 1780
1770 E1=1 @ GOTO 2050
1780 IF Y<10 THEN O1$=O1$&VAL$(Y) @ GOTO 1850
1790 IF Y=10 THEN O1$=O1$&"-" @ GOTO 1850
1800 IF Y=11 THEN O1$=O1$&"." @ GOTO 1850
1810 IF Y=12 THEN O1$=O1$&"9" @ GOTO 1850
1820 IF Y=14 THEN O1$=O1$&" " @ GOTO 1850
1830 IF Y=15 THEN O1$=O1$&"" @ GOTO 1850
1840 E1=1 @ GOTO 2050
1850 NEXT J
1860 !
1870 ! FORMATTING ERROR CHECK
1880 !
1890 IF C=1 OR C=P1 THEN 2060
1900 IF O1$(2,2)=" " THEN E1=1 @ GOTO 2050
1910 IF O1$(3,3)="#" THEN E1=1 @ GOTO 2050
1920 IF O1$(4,4)=" " THEN E1=1 @ GOTO 2050
1930 IF O1$(5,5)=" " THEN E1=1 @ GOTO 2050
1940 IF O1$(6,6)=" " THEN E1=1 @ GOTO 2050
1950 IF O1$(7,7)=" " THEN E1=1 @ GOTO 2050
1960 IF O1$(8,8)="#" THEN E1=1 @ GOTO 2050
1970 IF O1$(9,9)=" " THEN E1=1 @ GOTO 2050
1980 IF O1$(10,10)="#" THEN E1=1 @ GOTO 2050
1990 IF O1$(11,11)=" " THEN E1=1 @ GOTO 2050
2000 IF O1$(12,12)=" " THEN E1=1 @ GOTO 2050
2010 IF O1$(13,13)=" " THEN E1=1 @ GOTO 2050
2020 IF O1$(13,13)="-" THEN E1=1 @ GOTO 2050
2030 IF O1$(14,14)="#" " AND O1$(14,14)="#" THEN E1=1 @ GOTO
2050
2040 GOTO 2060

```

```

2050 IF E2<5 THEN E1=0 @ E2=E2+1 @ GOSUB 730 @ GOTO 1240
2060 DISP 01$
2070 SLET 02$(C) = 01$
2080 NEXT K
2090 !
2100 ! HP TAPE SETUP
2110 !
2120 IF F=0 THEN 2250
2130 N$=TRIM$(NO$)&VAL$(I)
2140 CREATE N$,3
2150 ASSIGN# 1 TO N$
2160 CRT OFF
2170 PRINT# 1 ; A$,B$,C$
2180 FOR K=1 TO P1
2190 01$=GET$(02$(K))
2200 PRINT# 1 ; 01$
2210 NEXT K
2220 ASSIGN# 1 TO *
2230 CRT ON
2240 !
2250 ! OUTPUT I/O INFORMATION
2260 !
2270 CRT OFF
2280 PRINT
2290 PRINT "HEADER = ";GET$(02$(1))
2300 IF E0>0 THEN PRINT "TRANSMIT ERRORS, FILE READ =
"&VAL$(E0)
2310 IF E2>0 THEN PRINT "FORMAT ERRORS, FILE READ =
"&VAL$(E2)
2320 IF E1>0 THEN PRINT "COMPLETED READ OF FILE ";I
2330 IF E1>0 THEN PRINT "WITH CHARACTER ERRORS"
2340 IF E1=0 THEN PRINT "SUCCESSFUL READ AND CONVERSION OF
FILE NUMBER ";I
2350 IF F=1 THEN PRINT "DATA OUTPUT TO TAPE FILE ";N$
2360 CRT ON
2370 NEXT I
2380 CLEAR
2390 PRINT
2400 PRINT "SELECTED NUMBER OF FILES READ!"
2410 DISP "SELECTED NUMBER OF FILES READ!"
2420 DISP "PRESS CONTINUE TO RESTART" @ PAUSE
2430 RETURN
2440 !
2450 ! END OF DATA ON TAPE
2460 !
2470 CLEAR
2480 PRINT
2490 PRINT "END OF DATA ON SIROTEM TAPE!"
2500 DISP "END OF DATA ON SIROTEM TAPE!"
2510 DISP "PRESS CONTINUE TO RESTART" @ PAUSE
2520 GOTO 460 ! RESTART
2530 !

```

```

2540 ! HELP ROUTINE
2550 !
2560 CLEAR @ CFLAG 1
2570 DISP "PRINT OR DISP INFORMATION? P/D  DEFAULT IS DISP"
2580 INPUT Z$
2590 IF Z$="" THEN Z$[1,1]="D" @ GOTO 2610
2600 IF UPC$(Z$[1,1])="P" THEN CRT IS 2 ELSE CRT IS 1
2610 CLEAR
2620 DISP "NOTE: IF THE HP-85 HANGS UP"
2630 DISP "DURING A DATA TRANSFER, TRY "
2640 DISP "PAUSING THE PROGRAM AND THEN"
2650 DISP "TOGGLE THE DUMP SWITCH ON THE"
2660 DISP "SIROTEM. THIS SHOULD SEND A"
2670 DISP "CHARACTER STRING TO THE HP-85"
2680 DISP "AND FINISH THE TRANSFER. IF ALL"
2690 DISP "ELSE FAILS TURN THE HP-85 OFF  AND START OVER."
2700 DISP
2710 IF UPC$(Z$[1,1])="D" THEN DISP "(PRESS CONTINUE)" @
PAUSE
2720 CLEAR
2730 DISP "PROBLEMS SOMETIMES OCCUR WHEN"
2740 DISP "READING THE FIRST SIROTEM TAPE"
2750 DISP "FILE. REPOSITIONING THE SIROTEM"
2760 DISP "TAPE AND RESTARTING THE PROGRAM SHOULD HELP."
2770 DISP
2780 IF UPC$(Z$[1,1])="D" THEN DISP "(PRESS CONTINUE)" @
PAUSE
2790 CLEAR @ KEY LABEL
2800 DISP "THE SPECIAL FUNCTION KEYS ARE"
2810 DISP "LABELED AT THE BOTTOM OF THE"
2820 DISP "HP-85 CRT SCREEN. UPPER LABELS"
2830 DISP "REFER TO SHIFTED KEYS AND"
2840 DISP "LOWER LABELS REFER TO UNSHIFTED"
2850 DISP "KEYS. EACH KEY, WHEN PRESSED,"
2860 DISP "WILL PERFORM THE FOLLOWING          DEFINED
FUNCTIONS:"
2870 DISP
2880 IF UPC$(Z$[1,1])="D" THEN DISP "(PRESS CONTINUE)" @
PAUSE
2890 CLEAR @ KEY LABEL
2900 DISP "REW - REWINDS THE SIROTEM TAPE."
2910 DISP
2920 DISP "FWD - ADVANCES THE SIROTEM TAPE ONE FILE."
2930 DISP
2940 DISP "REV - REVERSES THE SIROTEM TAPE ONE FILE."
2950 DISP
2960 IF UPC$(Z$[1,1])="D" THEN DISP "(PRESS CONTINUE)" @
PAUSE
2970 CLEAR @ KEY LABEL
2980 DISP "READ - INITIATES THE TAPE READ  SUBROUTINE."
2990 DISP
3000 DISP "HELP - TYPES OUT A BRIEF HELP  FILE."

```



```
3010 DISP
3020 IF UPC$(Z$(1,1))="D" THEN DISP "(PRESS CONTINUE)" @
PAUSE
3030 CRT IS 1
3040 RETURN
```